
OxCal code used in Bayesian models presented in this paper (those for Figures 8-14 and Table 2, main text).

Roundtop

Options()
 {
  Resolution=1;
  Curve="intcal20.14c";
  kIterations=3000;
 };
 Plot()
 {
  Outlier_Model("General",T(5),U(0,4),"t");
  Outlier_Model("Charcoal",Exp(1,-10,0),U(0,3),"t");
  Sequence()
  {
   Boundary("Start Roundtop");
   Phase("Roundtop")
   {
    Sequence()
    {
     Boundary("Start Early");
     Phase("Early")
     {
      R_Date("Y-1534,charcoal",880,60)
      {
       Outlier("Charcoal",1);
      };
      R_Date("AA-26541, maize",830,45)
      {
       Outlier("General",0.05);
      };
      Interval("Interval Early");
      Date("Date Early");
     };
     Boundary("End Early");
    };
    Sequence("House 1 to House 2")
    {
     Boundary("Start House 1");
     Phase("House 1")
     {
      R_Date("Beta-135880, charcoal",630,60)
      {
       Outlier("Charcoal",1);
      };
      R_Date("Beta-13588, charcoal",650,40)
      {
       Outlier("Charcoal",1);
      };
      R_Date("AA-21979, maize",675,55)
      {
       Outlier("Charcoal",1);
      };
      R_Date("AA-21980, twig",670,55)
      {
       Outlier("General",0.05);
      };
      R_Date("AA-23106, bean",660,50)
      {
       Outlier("Charcoal",1);
      };
      Interval("Interval House 1");
      Date("Date House 1");
     };
     Boundary("End House 1");
     Interval("Time Period House 1 to House 2");
     Boundary("Start House 2");
     Phase("House 2")
     {
      R_Date("Beta-135879, charcoal",300,80)
      {
       Outlier("Charcoal",1);
      };
      R_Date("AA-21978, maize",330,40)
      {
       Outlier("General",0.05);
      };
      Date("Date House 2");
      Interval("Interval House 2",U(80));
     };
     Boundary("End House 2");
    };
    R_Date("AA-26539, maize",440,45)
    {
     Outlier("General",0.05);
    };
    R_Date("AA-26540, bean",315,45)
    {
     Outlier("General",0.05);
    };
    Date("Date estimate Roundtop");
    Interval("Interval Roundtop");
   };
   Boundary("End Roundtop");
  };
  Difference("Time Period Early to House 1","Start House 1","End Early");
 };


The alternative version of the Roundtop model changes the Interval queries for House 1 and House 2 from the above code to:

Interval("Interval House 1",U(0,50));
Interval("Interval House 2",U(0,50));


Maxon-Derby

Options()
 {
  Resolution=1;
  Curve="intcal20.14c";
  kIterations=3000;
 };
 Plot()
 {
  Outlier_Model("General",T(5),U(0,4),"t");
  Outlier_Model("SSimple",N(0,2),0,"s");
  Outlier_Model("Charcoal",Exp(1,-10,0),U(0,3),"t");
  Sequence()
  {
   Boundary("Start Maxon-Derby");
   Phase("Maxon-Derby")
   {
    R_Date("Beta-142122", 890,40)
    {
     Outlier("Charcoal", 1);
    };
    R_Date("Beta-135881", 1000,50)
    {
     Outlier("Charcoal", 1);
    };
    R_Date("Beta-135889", 980,40)
    {
     Outlier("Charcoal", 1);
    };
    R_Date("Beta-135883", 840,70)
    {
     Outlier("Charcoal", 1);
    };
    R_Date("Beta-135882", 820,80)
    {
     Outlier("Charcoal", 1);
    };
    Sequence()
    {
     Boundary();
     Phase("House A")
     {
      R_Date("Beta-135881 House A F23", 1000,50)
      {
       Outlier("Charcoal", 1);
      };
      R_Date("UGAMS-53047, maize F14",741,20)
      {
       Outlier("General",0.05);
      };
      R_Date("UGAMS-53047r, maize F14",829,25)
      {
       Outlier("General",0.05);
      };
      R_Date("UGAMS-53048 House A near F14, maize",777,21)
      {
       Outlier("General",0.05);
      };
      R_Date("UGAMS-53049 House A, maize F23",748,21)
      {
       Outlier("General",0.05);
      };
      Interval("Interval House A");
      Date("Date Estimate House A");
     };
     Boundary();
    };
    Sequence()
    {
     Boundary();
     Phase("House C")
     {
      D_Sequence("Maxon-Derby WM")
      {
       Outlier("General",0.05);
       R_Date("Maxon7 RY1-9 UGAMS-59369",960,25)
       {
        Outlier("SSimple",0.05);
       };
       Gap(13);
       R_Date("Maxon7 RY16-20, UGAMS-59370",920,25)
       {
        Outlier("SSimple",0.05);
       };
       Gap(2);
       Date("Maxon7 Bark");
      };
      Interval("Interval House C");
      Date("Date Estimate House C");
     };
     Boundary();
    };
    Sequence()
    {
     Boundary();
     Phase ("House D")
     {
      R_Date("UGAMS-53050, F73 small pit, nutshell",806,22)
      {
       Outlier("General",0.05);
      };
      Interval("Interval House D");
      Date("Date Estimate House D");
     };
     Boundary();
    };
    Interval("Interval Maxon-Derby Overall");
    Date("Date Estimate Maxon-Derby");
   };
   Boundary("End Maxon-Derby");
  };
 };


Bates

Options()
 {
  Resolution=1;
  Curve="intcal20.14c";
  kIterations=3000;
 };
Plot()
 {
  Outlier_Model("General",T(5),U(0,4),"t");
  Outlier_Model("SSimple",N(0,2),0,"s");
  Outlier_Model("Charcoal",Exp(1,-10,0),U(0,3),"t");
  Sequence()
  {
   Boundary("Start Bates Early");
   Phase("Bates Early")
   {
    D_Sequence("Bates8")
    {
     Outlier("General",0.05);
     R_Date("Bates8 RY1-5 UGAMS-59367",890,20)
     {
      Outlier("SSimple",0.05);
     };
     Gap(25);
     R_Date("Bates8 RY26-30, UGAMS-59368",910,20)
     {
      Outlier("SSimple",0.05);
     };
     Gap(2);
     Date("Bates8 Bark");
    };
    D_Sequence("Bates1")
    {
     Outlier("General",0.05);
     R_Date("Bates1 RY1-5 UGAMS-59365",870,20)
     {
      Outlier("SSimple",0.05);
     };
     Gap(17);
     R_Date("Bates1 RY18-22, UGAMS-59368",820,25)
     {
      Outlier("SSimple",0.05);
     };
     Gap(2);
     Date("Bates1 Bark");
    };
    R_Date("Beta-135885", 910,50)
    {
     Outlier("Charcoal",1);
    };
    R_Date("Beta-135887", 890,60)
    {
     Outlier("Charcoal",1);
    };
    Date("Date Estimate Bates Early");
    Interval("Interval Bates Early");
   };
   Boundary("End Bates Early");
   Interval("Period of Time Between Bates Early and Bates Later");
   Boundary("Start Bates Later");
   Phase("Bates Later")
   {
    R_Date("Beta-135886", 720,60)
    {
     Outlier("Charcoal",1);
    };
    R_Date("UGAMS-53045, maize",670,21)
    {
     Outlier("General",0.05);
    };
    R_Date("UGAMS-53044, nutshell",655,21)
    {
     Outlier("General",0.05);
    };
    R_Date("UGAMS-53046r, maize",637,20)
    {
     Outlier("General",0.05);
    };
    Interval("Interval Bates Later");
    Date("Date Estimate Bates Later");
   };
   Boundary("End Bates Later");
  };
  Difference("Time Span Bates Overall","End Bates Later","Start Bates Early");
 };


Sackett

Options()
 {
  Resolution=1;
  Curve="intcal20.14c";
  kIterations=3000;
 };
 Plot()
 {
  Outlier_Model("General",T(5),U(0,4),"t");
  Outlier_Model("Charcoal",Exp(1,-10,0),U(0,3),"t");
  Sequence()
  {
   Boundary("Start Sackett");
   Phase("Sackett")
   {
    R_Date("M-1076_charcoal",820,75)
    {
     Outlier("Charcoal",1);
    };
    R_Date("Beta-135884_charcoal",750,50)
    {
     Outlier("Charcoal",1);
    };
    R_Date("Beta-135880_charcoal",840,40)
    {
     Outlier("Charcoal",1);
    };
    R_Date("UCIAMS-270843",635,20)
    {
     Outlier("General",0.05);
    };
    R_Date("UCIAMS-270844",675,20)
    {
     Outlier("General",0.05);
    };
    R_Date("UCIAMS-270845",590,20)
    {
     Outlier("General",0.05);
    };
    Date("Date estimate Sackett");
    Interval("Interval estimate Sackett");
   };
   Boundary("End Sackett");
  };
 };


Kelso

Options()
 {
  Resolution=1;
  Curve="intcal20.14c";
  kIterations=3000;
 }; 
 Plot("Kelso Site")
 {
  Outlier_Model("Charcoal",Exp(1,-10,0),U(0,3),"t");
  Outlier_Model("General",T(5),U(0,4),"t");
  Outlier_Model("SSimple",N(0,2),0,"s");
  Sequence()
  {
   Boundary("Start Kelso");
   Phase("Kelso")
   {
    R_Date("Beta-138610 charcoal",880,40)
    {
     Outlier("Charcoal",1);
    };
    R_Date("Beta-135891 charcoal",720,40)
    {
     Outlier("Charcoal",1);
    };
    R_Date("ISGS-A0657 maize",600,30)
    {
     Outlier("General",0.05);
    };
    R_Date("ISGS-A0661 maize",560,25)
    {
     Outlier("General",0.05);
    };
    R_Date("ISGS-A0660 grass",560,30)
    {
     Outlier("General",0.05);
    };
    R_Date("ISGS-A0759 residue",520,30)
    {
     Outlier("General",0.05);
    };
    R_Date("GrM-14982, maize",545,20)
    {
     Outlier("General",0.05);
    };
    R_Combine("Kelso_5")
    {
     Outlier("General",0.05);
     R_Date("UGAMS-35644_KELSO-5A_maize-split", 576,19)
     {
      Outlier("SSimple",0.05);
     };
     R_Date("GrM-14983_KELSO-5B_maize-split", 624,25)
     {
      Outlier("SSimple",0.05);
     };
    };
    Interval("Interval Kelso");
    Date("Date estimate Kelso");
   };
   Boundary("End Kelso");
  };
 };


The OxCal runfile for the alternative Kelso results and the longer settlement series shown in Figure 14 can be found in Birch et al. (2021) with the difference that for those sites dating from the mid-15th century and older (Bloody Hill and older in the Onondaga Sequence), we change the assumed constraint on an Interval query for the site duration from the later period assumption of LnN(ln(20),ln(2)) (see Manning et al. 2020; Birch et al. 2021) to instead a potentially much longer assumption of LnN(ln(40),ln(2)).

Supplementary Figure 1. Comparison of a KDE Plot query applied to the Sackett Phase (see Figures 11 and 14) versus a Date query applied to the same Sackett Phase. The modelled calendar probabilities are very similar. 

Supplementary Table 1. The results from the models reported in the main text in Table 2 except now with use of the Charcoal Plus Outlier model (Dee and Bronk Ramsey 2014). Gray shading indicates U(0,50) constraint applied to Interval query for Roundtop House 1 and House 2 (see OxCal runfiles above).
	Site
	Previous Estimates
	Bayesian Modeling Results 

	
	Ritchie and Funk 1973 
	Hart 2000
	Start Boundary
	Date Estimate
	End Boundary
	Agreement

	Roundtop (overall)
	
	
	1041-1042 (0.2)
1044-1228 (68.1)
1055-1057 (0.5)
1059-1231 (67.8)
	1190-1570 (68.3)
1196-1553 (68.3)
	1534-1715 (68.3)
1522-1689 (68.3)
	Model 128
Overall 124.3

Model 130.7
Overall 126.7

	
	
	
	774-1262 (95.4)
805-1243 (95.4)
	967-1782 (95.4)
986-1747 (95.4)
	1490-1965 (95.4)
1482-1907 (95.3)
1908-1914 (0.2)
	

	Roundtop Early
	1000-1100
	
	1152-1249 (68.3)
1155-1249 (68.3)
	1176-1272 (68.3)
1178-1272 (68.3)
	1195-1307 (68.3)
1197-1307 (68.3)
	

	
	
	
	1035-1270 (95.4)
1042-1270 (95.4)
	1068-1383 (95.4)
1077-1080 (0.1)
1081-1382 (95.4)
	1161-1514 (95.4)
1162-1500 (95.4)
	

	Difference Between
	
	
	4-153 or 23-160 years (68.3)
-181 to 206 or -152 to 209 years (95.4)
	

	Roundtop House 1
	
	1286-1392
	1281-1308 (21.5)
1347-1385 (46.8)
1292-1301 (7.0)
1351-1386 (61.3)
	1299-1318 (12.8)
1356-1397 (55.4)
1300-1309 (6.9)
1360-1395 (61.4)
	1317-1324 (3.5)
1366-1416 (64.8)
1315-1323 (5.8)
1369-1405 (62.4)
	

	
	
	
	1259-1392 (95.4)
1276-1321 (25.1)
1336-1393 (70.4)
	1282-1412 (95.4)
1285-1332 (24.9)
1346-1405 (70.6)
	1297-1441 (95.4)
1295-1343 (24.7)
1360-1416 (70.8)
	

	Interval Between
	
	
	38-156 or 73-186 years (68.3)
0-215 or 45-259 years (95.4)
	

	Roundtop House 2

	
	1453-1637
	1435-1535 (68.3)
1456-1541 (68.3)
	1464-1562 (68.3)
1468-1555 (68.3)
	1489-1594 (68.3)
1480-1565 (68.3)

	

	
	
	
	1385-1596 (95.4)
1438-1609 (95.4)
	1429-1636 (95.4)
1450-1620 (95.4)

	1462-1669 (95.4)
1461-1629 (95.4)
	

	Maxon-Derby
(overall)
	1100
	1082-1129 (early)
1214-1393 (late)
	1068-1154 (68.3)
	1142-1280 (68.3)
	1273-1328 (68.3)
	Model 99
Overall 95.2


	
	
	
	980-1168 (93.8)
1196-1213 (1.7)
	1058-1334 (95.4)
	1235-1391 (95.4)
	

	Maxon-Derby House A
	
	1041-1257
	1223-1269 (68.3)
	1232-1241 (9.2)
1249-1279 (59.1)
	1268-1290 (68.3)

	

	
	
	
	1152-1275 (95.4)
	1197-1295 (95.4)
	1232-1255 (9.8)
1259-1309 (85.6)
	

	Maxon-Derby House C
	
	
	1110-1166 (68.3)
	1119-1189 (68.3)
	1135-1223 (68.3)
	

	
	
	
	1037-1175 (93.7) 
1204-1216 (1.8)
	1059-1242 (95.4)
	1096-1303 (95.4)
	

	Maxon-Derby House D
	
	
	1176-1253 (68.3)
	1211-1269 (68.3)
	1233-1282 (68.3)
	

	
	
	
	1102-1267 (95.4)
	1146-1299 (95.4)
	1217-1332 (95.4)
	

	Bates Early
	1190
	1022-1257
	1161-1209 (68.3)
1071-1221 (95.4)
	1183-1228 (68.3)
1129-1268 (95.4)
	1200-1247 (68.3)
1187-1288 (95.4)
	Model 76.4
Overall 77.5

	Interval Between
	
	
	13-100 years (68.3); 0-150 years (95.4)
	

	Bates Later
	
	1214-1393
	1262-1306 (48.1)
1343-1372 (20.2)
	1294-1337 (33.3)
1351-1396 (35.0)
	1329-1345 (13.3)
1372-1428 (54.0)
	

	
	
	
	1242-1378 (95.4)
	1272-1428 (95.4)
	1301-1461 (95.4)
	

	Sackett
	1130
	1164-1282
	1207-1307 (68.3)
	1269-1368 (68.3)
	1313-1356 (36.3)
1362-1403 (32.0)
	Model 107.3
Overall 107.4

	
	
	
	1115-1370 (95.4)
	1185-1426 (95.4)
	1303-1477 (95.4)
	

	Kelso
	1390
	1045-1278

1325-1418 (Hart & Lovis 2007)
	1318-1337 (35.3)
1385-1403 (33.0)
	1326-1347 (34.2)
1393-1410 (34.0)
	1332-1353 (29.7)
1400-1423 (38.6)
	Model 91.1
Overall 91

	
	
	
	1279-1354 (53.8)
1364-1407 (41.7)
	1315-1427 (95.4)
	1327-1370 (40.5)
1396-1454 (54.9)
	

	Kelso alternative
	
	
	1388-1399 (68.3)
1379-1405 (95.4)
	1395-1411 (68.3)
1386-1425 (95.4)
	1405-1420 (68.3)
1400-1434 (95.4)

